

Lab 4: Linear Systems

This Maple lab is based in part on earlier versions prepared by Professors R. Falk and R. Bumby of the Rutgers Mathematics department.

Introduction. In this lab we use Maple to find *eigenvalues* and *eigenvectors* of matrices, and solve systems of linear *algebraic* equations as well as systems of first-order linear *differential* equations. We also obtain pictures of the slope fields of systems of two linear DE in the *phase plane*.

Please obtain the seed file from the web page and save it in your directory on eden. Then prepare the Maple lab according to the instructions and hints in the introduction to Lab 0. Turn in only the printout of your Maple worksheet, after removing any extraneous material and any errors you have made. For Lab4 you will also need to obtain a *supplementary worksheet* that contains some commands which you are asked to try out but which should not appear in the final lab.

0. Setup. As usual, the seed file begins with commands which load the required Maple packages: `with(plots):` and `with(DEtools):`. In this lab we also require a Linear Algebra package, which is loaded using the command `with(LinearAlgebra):`.

1. Matrix entry. The seed file and the supplementary worksheet include the definitions of three matrices, A , B and C :

```
A:= <<3|2|12>,<-4|9|14>>;
B:= <<3,4,6>|<2,3,-4>>;
C:= <<10,0>|<-2,6>|<-7,6>>;
```

The syntax for entering matrices into Maple should be clear from these examples (and the resulting matrices); if not, you can read about it by entering `?<` at the prompt. For practice, enter in your worksheet (i) a column vector with three entries, all different; (ii) a row vector with four entries, all different; and (iii) a 3×3 matrix whose entries are all distinct.

2. Matrix Operations. The goal of this section is to understand how the matrix operations of addition (+), matrix multiplication (.), scalar multiplication (*), and powers (^) act in various circumstances. We will experiment with these, and since some of our experiments may give errors and unexpected results, we will do so in the *supplementary worksheet*. The worksheet that you submit should contain only a *discussion*, guided by the questions below.

A few examples of the use of matrix operations are already in the supplementary worksheet, and you should add others to allow a full discussion of these operations. Some of these examples will lead to errors, and such errors will find their place in the worksheet discussion: your comment should include a description of the failed command and your interpretation of the error message. If you have any doubt about your interpretation of a result, you can consult *Maple help*. There are various ways to do so (see the *Help* button at the top of the worksheet) but here are two particular ones that may be helpful: at a command prompt `>`, typing “?anything” (`> ?anything`) will give help on the topic “anything” (try `> ?LinearAlgebra` or `> ?.`), and placing the cursor on a command you have typed, and pressing F2, will give help on that command.

Here are the questions to guide your discussion:

- (1) How is $M1+M2$ computed when $M1$ and $M2$ are matrices? Is it *always defined*? If not, how does Maple indicate that it is not defined?
- (2) How is $M1.M2$ computed when $M1$ and $M2$ are matrices? Is it *always defined*? If not, how does Maple indicate that it is not defined?
- (3) How is M^c computed when M is a matrix and c is an integer constant (possibly negative)? Is it *always defined*? If not, under what circumstances is it undefined, and how does Maple indicate the nature of the problem?

- (4) How are $\mathbf{c} \cdot \mathbf{M}$ and $\mathbf{M} \cdot \mathbf{c}$ computed when \mathbf{M} is a matrix and \mathbf{c} is a scalar? What problems can arise when one tries to carry out scalar multiplication using $\mathbf{c} \cdot \mathbf{M}$ or $\mathbf{M} \cdot \mathbf{c}$? Are these products always defined? Do they always give matrices as answers? Does it make any difference whether the scalar comes first or second in the product? Whether the scalar is a number (constant) or a variable?

3. Eigenvalues, eigenvectors, and matrix exponentials.

3a. Eigenvalues and eigenvectors. The following lines appear in the seed file. They use the `Eigenvectors` command to obtain eigenvectors *and* eigenvalues of three related matrices.

```
M1:=A.B;
M2:=B.A;
M3:=M1 ^ (-1);
(Vals1,Vecs1):=Eigenvectors(M1);
(Vals2,Vecs2):=Eigenvectors(M2);
(Vals3,Vecs3):=Eigenvectors(M3);
```

(1) What is the relation between $\mathbf{M1}$ and $\mathbf{M2}$? What is the relation between their eigenvalues? *Note:* The relation between the eigenvalues of $\mathbf{M1}=\mathbf{A} \cdot \mathbf{B}$ and $\mathbf{M2}=\mathbf{B} \cdot \mathbf{A}$ is an example of a general theorem, which we do not give.

(2) What is the relation between $\mathbf{M1}$ and $\mathbf{M3}$? What is the relation between their eigenvectors? Between their eigenvalues? Give a brief discussion (using the fact that $\mathbf{M}\mathbf{M}^{-1} = \mathbf{M}^{-1}\mathbf{M} = \mathbf{I}$) to explain why these relations should be true.

3b. Solution of Initial Value Problems for First Order Systems. The general solution of the system of differential equations presented in the form $\mathbf{y}' = \mathbf{A}\mathbf{y}$ can be written in the form $\mathbf{y} = \sum_{i=1}^n c_i e^{\lambda_i t} \mathbf{y}^i$, where c_i are arbitrary constants and $\lambda_1, \dots, \lambda_n$ and $\mathbf{y}^1, \dots, \mathbf{y}^n$ are the eigenvalues and corresponding eigenvectors of the matrix \mathbf{A} , assuming that $\lambda_1, \dots, \lambda_n$ are distinct. To satisfy the initial condition $\mathbf{y}(0) = \mathbf{y}^0$, we solve the linear system $\mathbf{\Psi}\mathbf{c} = \mathbf{y}^0$, where $\mathbf{\Psi}$ is the (*fundamental*) matrix whose columns are the eigenvectors $\mathbf{y}^1, \dots, \mathbf{y}^n$ and \mathbf{c} is a column vector with entries c_1, \dots, c_n .

The following Maple commands, included in your worksheet, solve for \mathbf{c} for the problem when \mathbf{A} is the matrix $\mathbf{M2}$ of part (a) and $\mathbf{y}(0) = (2, -1, 3)^T$ and compute the diagonal matrix $\mathbf{MVals2}$, whose diagonal entries are $e^{\lambda_1 t}, e^{\lambda_2 t}, e^{\lambda_3 t}$, where the λ_i are the eigenvalues $\mathbf{Vals2}$ of the matrix $\mathbf{M2}$.

```
c := LinearSolve(Vecs2, <2, -1, 3>);
MVals2:= <<exp(Vals2[1]*t)|0|0>, <0|exp(Vals2[2]*t)|0>, <0|0|exp(Vals2[3]*t)>>;
```

Enter into your Maple worksheet a formula that computes the solution $\mathbf{Y1}$ of this initial value problem using the vector \mathbf{c} and the matrices $\mathbf{MVals2}$ and $\mathbf{Vecs2}$.

3c. Solution Using Matrix Exponentials. A simpler way to find explicit solutions of initial value problems associated to differential equations presented in the form $\mathbf{y}' = \mathbf{A}\mathbf{y}$ is to use the matrix exponential function $e^{\mathbf{A}t}$. As noted in Boyce and diPrima, Section 7.7, if $\mathbf{x}^1, \dots, \mathbf{x}^n$ are the special fundamental set of solutions of the equation $\mathbf{y}' = \mathbf{A}\mathbf{y}$ satisfying the initial conditions of Theorem 7.4.4, then $e^{\mathbf{A}t}$ is the same as the *fundamental matrix* $\Phi(t)$, whose columns are the vectors $\mathbf{x}^1, \dots, \mathbf{x}^n$. The matrix exponential $e^{\mathbf{A}t}$ then satisfies the key equations $d/dt(e^{\mathbf{A}t}) = \mathbf{A}e^{\mathbf{A}t}$ and $e^{\mathbf{A}t}|_{t=0} = \mathbf{I}$. Hence, the solution of the Initial Value Problem $\mathbf{y}' = \mathbf{A}\mathbf{y}$, $\mathbf{y}(0) = \mathbf{y}^0$, is given by $\mathbf{y} = e^{\mathbf{A}t}\mathbf{y}^0$. The `LinearAlgebra` package provides this function for us and a sample command is included in the work sheet.

To differentiate the matrix exponential, or any matrix or vector function of t , we must use the maple command `map`. (*Warning:* There is also a `Map` function, and the two are not interchangeable.

We use only `map` in this lab.) If M is a Maple Matrix (or Vector) whose entries depend on the variable t , then the command `map(diff,M,t)` constructs a new Matrix (or Vector) whose entries are obtained by differentiating the entries of M .

The following Maple commands, included in the worksheet, construct the fundamental matrix and check that it has all the required properties:

```
E2:=MatrixExponential(M2,t);
DE2:=map(diff,E2,t); #derivative of the matrix exponential
ME2:=M2.E2;
DE2-ME2;# checks equation if zero matrix
subs(t=0,E2);# checks if initial value is identity matrix
```

To illustrate the connection between the matrix exponential and solutions of individual initial value problems, you should:

(1) use the matrix `E2` to find the solution `Y2` of

$$\frac{d\mathbf{y}}{dt} = M_2\mathbf{y} \quad \text{with} \quad \mathbf{y}(0) = \begin{pmatrix} 2 \\ -1 \\ 3 \end{pmatrix}$$

(2) check that this vector satisfies the differential equations; and check that this vector satisfies the initial conditions.

(3) check that `Y2` and the solution `Y1` obtained in (3b) are the same.

As discussed in Boyce and DiPrima Section 7.7 (but only partially), when A is an $n \times n$ matrix with n distinct eigenvalues $\lambda_1, \dots, \lambda_n$ and corresponding eigenvectors $\xi^{(1)}, \dots, \xi^{(n)}$, the matrix e^{At} can be obtained by the formula $e^{At} = TQ(t)T^{-1}$; here $Q(t)$ is a diagonal matrix whose i^{th} diagonal entry is $e^{\lambda_i t}$ and T is a matrix whose columns are the eigenvectors: $T = (\xi^{(1)} \dots \xi^{(n)})$. The worksheet includes the instructions

```
Q2 := MVals2; # note the matrix Q2 has been computed in 3b
E2a := Vecs2.Q2.Vecs2^(-1); # alternative formula for matrix exponential
E2-E2a; # checks alternate formula if zero
```

which construct the matrix in this way (when $A = M_2$) and then check that the constructed matrix agrees with the one Maple found with its built-in Matrix Exponential function. You should try out these instructions and understand what they are doing, but no questions are asked about them.

4. Saddle points, nodes, and spirals. Consider the matrices

$$M_{4A} = \begin{pmatrix} -4 & -3 \\ 2 & 3 \end{pmatrix}, \quad M_{4B} = \begin{pmatrix} -2 & 2 \\ -5 & 0 \end{pmatrix}, \quad \text{and} \quad M_{4C} = \begin{pmatrix} -1 & -2 \\ 6 & 6 \end{pmatrix}.$$

For each, we will use the matrix exponential to solve the equation $d\mathbf{y}/dt = M\mathbf{y}$ with initial conditions

$$(a) \mathbf{y}(0) = \begin{pmatrix} 1 \\ 0 \end{pmatrix}, \quad (b) \mathbf{y}(0) = \begin{pmatrix} -1 \\ 0 \end{pmatrix}, \quad (c) \mathbf{y}(0) = \begin{pmatrix} -1/2 \\ 1/2 \end{pmatrix}, \quad (d) \mathbf{y}(0) = \begin{pmatrix} 1/2 \\ -1/2 \end{pmatrix}.$$

We will also make a *graphical* check of the solution by plotting the slope field of the equation in the *phase plane*, whose coordinates are the components of \mathbf{y} , and superimposing a *parametric plot* of the *trajectories* of the solutions. Because the equations are *autonomous*, all solutions starting at

a point on one of these trajectories will follow the trajectory—the only difference being the value of t at which it visits a particular point.

4a. Maple instructions: matrix M_{4A} . Here are the instructions that appear in the seed file for getting graphs for the first system.

```
M4A:=<<-4,2>|<-3,-3>>;
(Vals4A,Vecs4A):=Eigenvectors(M4A);
E4A:=MatrixExponential(M4A,t);
Y4Aa:=E4A.<1,0>;
Y4Ab:=E4A.<-1,0>;
Y4Ac:=E4A.<-1/2,1/2>;
Y4Ad:=E4A.<1/2,-1/2>;
tvals:=t=-3..3;
yvals:=y1=-3..3,y2=-3..3;
eq4A:=[diff(y1(t),t)=(M4A.<y1(t),y2(t)>)[1],
diff(y2(t),t)=(M4A.<y1(t),y2(t)>)[2]];
Field4A:=DEplot(eq4A,[y1(t),y2(t)],tvals,yvals,color=GREEN);
Sol4A:=plot([[Y4Aa[1],Y4Aa[2],tvals],
[Y4Ab[1],Y4Ab[2],tvals],
[Y4Ac[1],Y4Ac[2],tvals],
[Y4Ad[1],Y4Ad[2],tvals]],
yvals,color=[BLACK,RED,BLUE,CYAN]);
display({Field4A,Sol4A},title="Equation 4A");
```

There are a couple of points in the above instructions where further explanation might be helpful.

- Since the `diff` operation in Maple only applies to scalar functions, we must define each equation separately. (The same effect could be obtained with `map`, but the approach here is a good alternative when there are only two entries.) The expression `(M4A.<y1(t),y2(t)>)[1]` forms a column vector from the two components $y_1(t)$ and $y_2(t)$ of the solution, multiplies this vector on the left by the matrix M_{4A} , and takes the first component of the resulting vector.
- The expression `Sol4A` plots a *list* of objects, each of which is a parametric description of the trajectory of a solution in the phase plane. Colors are assigned to the plots in the same order that they appear in the list.

4b, 4c. Maple instructions: matrices M_{4B} and M_{4C} .

4d. Discussion. After you have created the plots for each of M_{4A} , M_{4B} , and M_{4C} , you should discuss several aspects of the results. In particular:

- (1) How can you tell *from your plots* what the direction of flow along the solution curves is?
- (2) For each of the three equations:
 - (a) Identify *from the plot* the type of equilibrium point at the origin: saddle point, stable or unstable node or spiral. Explain briefly.
 - (b) Discuss whether the eigenvalues of the corresponding matrix (M_{4B} , and M_{4C}) support your conclusion in (a).

4e. Straight line trajectories.. Finally, for the *saddle point* example, construct a second graph by adding to the graph already constructed the four “special” trajectories which lie along straight lines. Hint: You will need the eigenvectors found above.

End of Lab 4